

# **PULSER (PLUS) PRIZMATIX**

SOF-732-6

USER'S MANUAL

**DOC-323**

**Rev. 1.0**

Copyright ©2019  
All Rights Reserved

Med Associates, Inc.  
P.O. Box 319  
St. Albans, Vermont 05478

Phone: 802.527.2343  
Fax: 802.527.5095  
[www.med-associates.com](http://www.med-associates.com)

**This page intentionally left blank**

## Table of Contents

<b>Chapter 1   Getting Started.....</b>	<b>1</b>
Introduction.....	1
Software Installation and Setup.....	1
Com Port Lookup.....	1
<b>Chapter 2   MEDSTATE NOTATION™ COMMANDS.....</b>	<b>3</b>
OpenPulser.....	3
SetPulserParams (Prizmatix Pulser ONLY).....	3
SetPulserPlusParams (Prizmatix Pulser Plus ONLY).....	3
Operation Mode.....	4
StopPulser.....	4
ClosePulser.....	4
<b>Chapter 3   Sample Code.....</b>	<b>5</b>
<b>Chapter 4   Contact Information.....</b>	<b>7</b>



## CHAPTER 1 | GETTING STARTED

### Introduction

The SOF-732-6 Pulser (Plus) Prizmatix software enables Med-PC® users to program trains of pulses, to create groups of trains, and to add various triggering conditions for pulsing LEDs, lasers and shutters used in Optogenetics experiments via Med-PC® protocols and Med Associates, Inc. hardware while simultaneously collecting experimental data.

The software includes a sample code that may be added to any protocol as well as a Fixed Ratio protocol.

---

**Note: Requires a TTL output from the Med Associates interface to operate at 1ms resolution.**

---

### Software Installation and Setup

Before beginning the software installation, phone, fax or e-mail Med Associates with the registration information.

Insert the **SOF-732-6 PULSER (PLUS) PRIZMATIX** CD into the CD drive; at the welcome screen click **Install**.

If the CD does not auto-start, navigate to the CD drive in Windows Explorer and double click “**autorun.exe**” and click **Install** at the welcome screen.

### Com Port Lookup

MED-PC requires the Serial Com Port number of the Pulser (Plus) to communicate. See Figure 1-1-Device Manager / Com Port Lookup.


Once the Pulser (Plus) is connected via USB:

- Open Windows Device Manager.
- Expand the Ports (COM & LPT) category.
- The Pulser (Plus) will be identified as a USB Serial Port (COM#).

If you have more than one device like this in the category:

- Right-click on USB Serial Port (COM#).
- Select Properties.
- Under the General tab, the Manufacturer will be FTDI.

Figure 1-1- Device Manager / Com Port Lookup



## CHAPTER 2 | MEDSTATE NOTATION™ COMMANDS

### OpenPulser

Syntax	<code>OpenPulser(MG: MPCGlobalPtr; BOX: Real; portNumber: Integer): Integer;</code>
Parameters	<b>MG:</b> The MED-PC Global Parameter. Used for passing error messages back to MED-PC. <b>BOX:</b> The MED-PC Box number that is running the program. <b>portNumber:</b> The Serial Com Port # of the Prizmatix Pulser (e.g. 3)
Comments	Opens and configures the Serial Com Port of the Prizmatix Pulser. OpenPulser = 1 if successful, = 0 if unsuccessful.
Example	<code>~M := OpenPulser(MG, BOX, Trunc(P[4]));~;</code> Sets the Variable M to the Value returned by OpenPulser.

### SetPulserParams (Prizmatix Pulser ONLY)

Syntax	<code>SetPulserParams(MG: MPCGlobalPtr; BOX: Real; Var P);</code>
Parameters	<b>MG:</b> The MED-PC Global Parameter. <b>BOX:</b> The MED-PC Box number that is running the program. <b>P:</b> The array with the Prizmatix Pulser parameters. The array must be declared with 4 [0,1,2,3] elements and does not have to be the array X in the MSN code. P[0] = Frequency (Hz) P[1] = Pulse Width (ms) <b>*Must be 1ms &lt; (1 / Frequency)</b> P[2] = Stimulus Duration (sec) P[3] = Operation Mode <b>*See Operation Mode</b>
Comments	Calculate and send the pulse parameters to the Prizmatix Pulser.
Example	<code>~SetPulserPlusParams(MG, BOX, P);~;</code> Sets the elements [0,1,2,3] of array P with defined values.

### SetPulserPlusParams (Prizmatix Pulser Plus ONLY)

Syntax	<code>SetPulserPlusParams(MG: MPCGlobalPtr; BOX: Real; Var P);</code>
Parameters	<b>MG:</b> The MED-PC Global Parameter. <b>BOX:</b> The MED-PC Box number that is running the program. <b>P:</b> The array with the Prizmatix Pulser parameters. The array must be declared with 4 [0,1,2,3] elements and does not have to be the array X in the MSN code. P[0] = Frequency (Hz) P[1] = Pulse Width (ms) <b>*Must be 1ms &lt; (1 / Frequency)</b> P[2] = Stimulus Duration (sec) P[3] = Operation Mode <b>*See Operation Mode</b>
Comments	Calculate and send the pulse parameters to the Prizmatix Pulser.
Example	<code>~SetPulserPlusParams(MG, BOX, P);~;</code> Sets the elements [0,1,2,3] of array P with defined values.

## Operation Mode

### Pulser

- 0 = Execute pulse sequence after parameters sent.
- 1 = Execute pulse sequence one time only after trigger, parameters must then be sent again.
- 2 = Execute pulse sequence each time after trigger.
- 3 = Execute pulse sequence after trigger HIGH, then stop when LOW.

### Pulser Plus

- 100 = Execute pulse sequence after parameters sent.
- 101 = Execute pulse sequence one time only after trigger, parameters must then be sent again.
- 102 = Execute pulse sequence each time after trigger.
- 103 = Execute pulse sequence after trigger HIGH, then stop when LOW.

### StopPulser

Syntax	<code>StopPulser(MG: MPCGlobalPtr; BOX: Real);</code>
Parameters	<b>MG:</b> The MED-PC Global Parameter. <b>BOX:</b> The MED-PC Box number that is running the program.
Comments	Stops the current operation on the Prizmatix Pulser.
Example	<code>~StopPulser(MG, BOX);~;</code>

### ClosePulser

Syntax	<code>ClosePulser(MG: MPCGlobalPtr; BOX: Real);</code>
Parameters	<b>MG:</b> The MED-PC Global Parameter. <b>BOX:</b> The MED-PC Box number that is running the program.
Comments	Closes the Serial Com Port of the Prizmatix Pulser.
Example	<code>~ ClosePulser(MG, BOX);~;</code>

**CHAPTER 3 | SAMPLE CODE**

```

\ Inputs
^LeftLever = 1
^RightLever = 2

\ Outputs
^LeftLever = 1
^RightLever = 2
^PulserTTL = 3

\ P() = Control Variables with Assigned Aliases as Defined
Var_Alias Frequency (Hz) = P(0) \ Default = 20 Hz
Var_Alias Pulse Width (ms) Must be lms < (1 / Frequency) = P(1) \ Default = 5 ms
Var_Alias Stimulus Duration (sec) = P(2) \ Default = 3 seconds
Var_Alias Operation Mode = P(3) \ Default = 2 (See below)
Var_Alias COM Port for Prizmatix Pulser = P(4) \ Default = 3

\ Constants for the Prizmatix Pulser
^Frequency = 0
^PulseWidth = 1
^Duration = 2
^OperationMode = 3
^COMPort = 4

\ M = Response from the Prizmatix Pulser Open Command
\ S = Elapsed Time in Session
\ Array for the Prizmatix Pulser parameters
DIM P = 4
\ P(^OperationMode) = 0 = Execute pulse sequence after params sent
\ 1 = Execute pulse sequence one time only after trigger, params must then
be sent again
\ 2 = Execute pulse sequence each time after trigger
\ 3 = Execute pulse sequence after trigger HIGH, then stop when LOW

\ K-Pulses Used in this Program
\ K1 - Trigger the Pulser.
\ K2 - End Program.

\ Z-Pulses Used in this Program
^Z_End = 32 \ Signal End of Session

DISKOLUMNS = 5

```

```

\*****
\
\          Pulser Sample
\ S1 - Set Default Values
\ Frequency                (20 Hz)
\ Pulse Width (Must be lms < (1 / Frequency)) (5 ms)
\ Stimulus Duration        (3 seconds)
\ Operation Mode           (2)
\ COM Port for Prizmatix Pulser (3)
\*****
S.S.1,
S1,
  0.01": SET P(^COMPort)      = 3, P(^Frequency)      = 20, P(^PulseWidth) = 5;
        SET P(^Duration)    = 3, P(^OperationMode) = 2;
        ---> S2

S2,    \ First Statement: Wait for START signal.
        \
        \ Second Statement: Update screen display with default values
        \ for Control Variables. This will show any changes made via
        \ the "Configure | Change Variables" Window prior to START.
        #START: SHOW 1,Session,S/60 ---> S3
        1": SHOW 6,COM Port,P(^COMPort),          7,Frequency,P(^Frequency),          8,Pulse
Width,P(^PulseWidth);
        SHOW 9,Stim Duration,P(^Duration), 10,Operation Mode,P(^OperationMode) ---> SX

S3,
  1": ADD S; SHOW 1,Session,S/60 ---> SX
  #Z^Z_End: ---> S2

\*****
\
\          PRIZMATIX PULSER CONTROL
\*****
S.S.2,
S1,    \ Open the Prizmatix Pulser
        \ Show error and exit if there was a problem
        #START: ~M := OpenPulser(MG, BOX, Trunc(P[4]));~;
        IF M = 1 [@Open, @NotOpen]
            @Open: SHOW 5,Pulser Opened,M ---> S2
            @Not:  SHOW 5,Pulser Not Opened,M;
                    Z^Z_End ---> SX

S2,    \ Send the stimulus parameters to the Prizmatix Pulser
        \ NOTE: At the 57600 Baud Rate for the COM Port it will
        \ take ~5ms (0.005s) to send the command to the Pulser
  0.01": ~SetPulserParams(MG, BOX, P);~ ---> S3

```

```
S3,      \ Wait for K-Pulse, then send TTL signal to the Pulser
        #K1: ON ^PulserTTL ----> S4
        #K2: Z^Z_End ----> S5

S4,
        0.01": OFF ^PulserTTL ----> S3
        #K2: Z^Z_End ----> S5

S5,      \ Wait for Screen Update and end with
        \ STOPABORTFLUSH for Automatic Data Saving
        2": ~ClosePulser(MG, BOX);~ ----> STOPABORTFLUSH
```

## CHAPTER 4 | CONTACT INFORMATION

Please contact Med Associates, Inc. for information regarding any of our products.

For Technical questions, email [support@med-associates.com](mailto:support@med-associates.com).

For Sales questions, email [sales@med-associates.com](mailto:sales@med-associates.com).

Visit our website at [www.med-associates.com](http://www.med-associates.com).